
Tesla

Tesla is an HTTP client that leverages middleware to streamline HTTP requests
and responses over a common interface for various adapters.
It simplifies HTTP communication by providing a flexible and composable
middleware stack. Developers can easily build custom API clients by stacking
middleware components that handle tasks like authentication, logging, and
retries. Tesla supports multiple HTTP adapters such as Mint, Finch,
Hackney, etc.
Tesla is ideal for developers who need a flexible and efficient HTTP client.
Its ability to swap out HTTP adapters and create custom middleware pipelines
empowers you to make different architectural decisions and build tools tailored
to your application's needs with minimal effort.
Inspired by Faraday from Ruby.

  
    
  
  Getting started


Add :tesla as dependency in mix.exs:
defp deps do
  [
     # or latest version
    {:tesla, "~> 1.11"},
    # optional, required by JSON middleware
    {:jason, "~> 1.4"},
    # optional, required by Mint adapter, recommended
    {:mint, "~> 1.0"}
  ]
end
:httpc as default Adapter
The default adapter is erlang's built-in httpc, primarily to avoid
additional
dependencies when using Tesla in a new project. But it is not recommended to
use it in production environment as it does not validate SSL certificates

License
    


The MIT License (MIT)

Copyright (c) 2015-Present Tymon Tobolski

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.

Client
    

In Tesla, a client is an entity that combines middleware and an adapter,
created using Tesla.client/2. Middleware components modify or enhance requests
and responses—such as adding headers or handling authentication—while adapters
handle the underlying HTTP communication. For more details, see the sections on
middleware and adapters.

  
    
  
  Creating a Client


A client is created using Tesla.client/2, which takes a list of middleware
and an adapter.
client = Tesla.client([Tesla.Middleware.PathParams, Tesla.Middleware.Logger])
You can then use the client to make requests:
Tesla.get(client, "/users/123")

  
    
  
  Passing Options to Middleware


You can pass options to middleware by registering the middleware as a tuple
of two elements, where the first element is the middleware module and the
second is the options.
client = Tesla.client(
  [{Tesla.Middleware.BaseUrl, "https://api.example.com"}]
)

  
    
  
  Passing Adapter


By default, the global adapter is used. You can override this by passing an
adapter to the client.
client = Tesla.client([], Tesla.Adapter.Mint)
You can also pass options to the adapter.
client = Tesla.client([], {Tesla.Adapter.Mint, pool: :my_pool})

  
    
  
  Single-Client (Singleton) Pattern


A common approach in applications is to encapsulate client creation within a
module or function that sets up standard middleware and adapter configurations.
This results in a single, shared client instance used throughout the
application. For example:
defmodule MyApp.ServiceName do
  defp client do
    middleware = [
      {Tesla.Middleware.BaseUrl, "https://api.service.com"},
      {Tesla.Middleware.BearerAuth, token: bearer_token()},
      # Additional middleware...
    ]
    Tesla.client(middleware, adapter())
  end

  defp adapter do
    Keyword.get(config(), :adapter)
  end

  defp bearer_token do
    Keyword.fetch!(config(), :bearer_token)
  end

  defp config do
    Application.get_env(:my_app, __MODULE__, [])
  end
end
In this pattern, the client is constructed internally, and operations use this
singleton client:
defmodule MyApp.ServiceName do
  def operation_name(body) do
    url = "/endpoint"
    # The client() function is called internally
    response = Tesla.post!(client(), url, body)
    # Process the response...
  end

  defp client do
    # Client construction as shown earlier
  end
end
You can then use the module to make requests without managing the client externally:
{:ok, response} = MyApp.ServiceName.operation_name(%{key: "value"})

  
    
  
  Multi-Client Pattern


In scenarios where different configurations are needed—such as multi-tenancy
applications or interacting with multiple services—you can modify the client
function to accept configuration parameters. This allows for the creation of
multiple clients with varying settings:
defmodule MyApp.ServiceName do
  def operation_name(client, body) do
    url = "/endpoint"
    # The client is passed as a parameter
    response = Tesla.post!(client, url, body)
    # Process the response...
  end

  def client(opts) do
    middleware = [
      {Tesla.Middleware.BaseUrl, opts[:base_url]},
      {Tesla.Middleware.BearerAuth, token: opts[:bearer_token]},
      # Additional middleware...
    ]
    Tesla.client(middleware, opts[:adapter])
  end
end
Now, you can create clients with different configurations:
client = MyApp.ServiceName.client(
  base_url: "https://api.service.com",
  bearer_token: "token_value",
  adapter: Tesla.Adapter.Hackney
  # Additional options...
)
{:ok, response} = MyApp.ServiceName.operation_name(client, %{key: "value"})

  
    
  
  Comparing Single-Client and Multi-Client Patterns


The choice between using a single-client (singleton) or multi-client pattern
depends on your specific needs:
	Library Authors: It's generally advisable to avoid the singleton client
pattern. Hardcoding configurations can limit flexibility and hinder users in
multi-tenant environments. Providing the ability to create clients with custom
configurations makes your library more adaptable and user-friendly.

	Application Developers: For simpler applications, a singleton client might
suffice initially. However, adopting the multi-client approach from the outset
can prevent future refactoring if your application grows or needs change.


Understanding these patterns helps you design applications and libraries that
are flexible and maintainable, aligning with best practices in software
development.

Testing
    

There are two primary ways to mock requests in Tesla:
	Using Mox
	Using Tesla.Mock (deprecated)

You can also create a custom mock adapter if needed. For more information about
adapters, refer to the Adapter Guide to create your own.

  
    
  
  Should I Use Mox or Tesla.Mock?


We recommend using Mox for mocking requests in tests because it
is well-established in the Elixir community and provides robust features for
concurrent testing. While Tesla.Mock offers useful capabilities, it may be
removed in future releases. Consider using Mox to ensure long-term
compatibility.
For additional context, see GitHub Issue #241.

  
    
  
  References


	How-To Test Using Mox



  

  
    
    Middleware - tesla v1.14.0
    
    

    


  
  

    
Middleware
    

TL;DR: adapter(middleware3(middleware2(middleware1(env, next, options))))
Middleware in Tesla extends the request/response pipeline. Requests pass
through a stack of middleware before reaching the adapter, allowing
modifications to both requests and responses.
Middleware can be a module implementing Tesla.Middleware
behaviour or a function matching Tesla.Middleware.call/3. There is no
distinction between request and response middleware; it's about when you
execute Tesla.run/2.
The middleware stack is processed by calling Tesla.run/2 until it reaches
the adapter.

  
    
  
  Writing Middleware


Example of a custom middleware module:
defmodule Tesla.Middleware.MyCustomMiddleware do
  @behaviour Tesla.Middleware

  @impl Tesla.Middleware
  def call(env, next, options) do
    # Actions before calling the next middleware
    # ...
    Tesla.run(env, next)
    # Actions after calling the next middleware
    # ...
  end
end
A request logger middleware example:
defmodule MyApp.Tesla.Middleware.Logger do
  require Logger

  @behaviour Tesla.Middleware

  def call(env, next, _) do
    Logger.info("Request: #{inspect(env)}")
    case Tesla.run(env, next) do
      {:ok, env} ->
        Logger.info("Response: #{inspect(env)}")
        {:ok, env}

      {:error, reason} ->
        Logger.error("Error: #{inspect(reason)}")
        {:error, reason}
    end
  end
end

  
    
  
  Production-Ready Middleware Pipeline Example


In a production application, you might combine built-in and custom middleware.
Here's an example pipeline:
defmodule MyApp.ServiceName do
  defp middleware do
    base_url = "..."
    token = "..."

    [
      # Preserve the original request, should be the first middleware in the
      # pipeline to preserve the original request.
      Tesla.Middleware.KeepRequest,
      # Preserve original path parameters, should be after KeepRequest
      # to preserve the original path parameters tokens
      Tesla.Middleware.PathParams,
      # Set the base URL
      {Tesla.Middleware.BaseUrl, base_url},
      # Add headers
      {Tesla.Middleware.Headers, [{"user-agent", "MyApp/1.0"}]},
      # Add authorization
      {Tesla.Middleware.BearerAuth, [token: token]},
      # Process the body (encoding, compression)
      # Use string keys for untrusted input
      Tesla.Middleware.JSON,
      # Compress the request and response
      #Tesla.Middleware.Compression,

      # ------------------------------------------------------------------------
      # Keep the telemetry and logging as close as possible to the actual request
      # being made.
      # Log requests and responses
      Tesla.Middleware.Logger,
      # Telemetry for timing
      {Tesla.Middleware.Telemetry},
      # Optional OpenTelemetry middleware
      # Tesla.Middleware.OpenTelemetry
    ]
  end
end
See built-in middlewares
for more examples.

Adapter
    

An adapter in Tesla implements the Tesla.Adapter behaviour and handles the
actual HTTP communication. It's the final step in the middleware chain,
responsible for sending requests and receiving responses.

  
    
  
  Writing an Adapter


You can create a custom adapter by implementing the Tesla.Adapter behaviour.
Here's an example:
defmodule Tesla.Adapter.Req do
  @behaviour Tesla.Adapter

  @impl Tesla.Adapter
  def run(env, _opts) do
    req = Req.new(
      url: Tesla.build_url(env),
      method: env.method,
      headers: env.headers,
      body: env.body
    )

    case Req.request(req) do
      {:ok, %Req.Response{} = resp} ->
        {:ok, %Tesla.Env{env | status: resp.status, headers: resp.headers, body: resp.body}}

      {:error, reason} ->
        {:error, reason}
    end
  end
end

  
    
  
  Setting the Adapter


If you don't specify an adapter when creating a client with Tesla.client/2,
Tesla uses the adapter configured in the :tesla application environment.
By default, Tesla uses Tesla.Adapter.Httpc, which relies on Erlang's built-in
httpc.
:httpc as default Adapter
The default httpc adapter is not recommended for production because it
doesn't validate SSL certificates and has other issues. Consider using Mint,
Finch, or Hackney adapters instead.

  
    
  
  Adapter Options


You can pass options to adapters in several ways:
	In the application configuration:
config :tesla, adapter: {Tesla.Adapter.Hackney, [recv_timeout: 30_000]}

	When creating a client:
defmodule MyService do
  def client(...) do
    middleware = [...]
    adapter = {Tesla.Adapter.Hackney, [recv_timeout: 30_000]}
    Tesla.client(middleware, adapter)
  end
end

	Directly in request functions:
Tesla.get(client, "/", opts: [adapter: [recv_timeout: 30_000]])



  
    
  
  About :httpc adapter and security issues


People have complained about :httpc adapter in Tesla due to
its security issues. The main problem is that :httpc does not validate SSL
certificates by default. Which, we believe, is a serious security issue and
should be addressed by :httpc itself.
As much as we would like to fix it, we can't, because we are unsure if it would
break existing code. We are not planning to fix it in Tesla due to backward
compatibility. We may reconsider this decision for a version 2.0.

Basic Cheat Sheet
    


  
    
  
  Making Requests 101



  
    
  
  Creating a client


client = Tesla.client([{Tesla.Middleware.BaseUrl, "https://httpbin.org"}])
Tesla.get(client, "/path")

  
    
  
  All Methods


Tesla.get("https://httpbin.org/get")

Tesla.head("https://httpbin.org/anything")
Tesla.options("https://httpbin.org/anything")
Tesla.trace("https://httpbin.org/anything")

Tesla.post("https://httpbin.org/post", "body")
Tesla.put("https://httpbin.org/put", "body")
Tesla.patch("https://httpbin.org/patch", "body")
Tesla.delete("https://httpbin.org/anything")

  
    
  
  Query Params


# GET /path?a=hi&b[]=1&b[]=2&b[]=3
Tesla.get("https://httpbin.org/anything", query: [a: "hi", b: [1, 2, 3]])

  
    
  
  Request Headers


Tesla.get("https://httpbin.org/headers", headers: [{"x-api-key", "1"}])

  
    
  
  Client Default Headers


client = Tesla.client([{Tesla.Middleware.Headers, [{"user-agent", "Tesla"}]}])

  
    
  
  Multipart


You can pass a Tesla.Multipart struct as the body:
alias Tesla.Multipart

mp =
  Multipart.new()
  |> Multipart.add_content_type_param("charset=utf-8")
  |> Multipart.add_field("field1", "foo")
  |> Multipart.add_field("field2", "bar",
    headers: [{"content-id", "1"}, {"content-type", "text/plain"}]
  )
  |> Multipart.add_file("test/tesla/multipart_test_file.sh")
  |> Multipart.add_file("test/tesla/multipart_test_file.sh", name: "foobar")
  |> Multipart.add_file_content("sample file content", "sample.txt")

{:ok, response} = Tesla.post("https://httpbin.org/post", mp)

  
    
  
  Streaming



  
    
  
  Streaming Request Body


If adapter supports it, you can pass a Stream
as request body, e.g.:
defmodule ElasticSearch do
  def index(records_stream) do
    stream = Stream.map(records_stream, fn record -> %{index: [some, data]} end)
    Tesla.post(client(), "/_bulk", stream)
  end

  defp client do
    Tesla.client([
      {Tesla.Middleware.BaseUrl, "http://localhost:9200"},
      Tesla.Middleware.JSON
    ], {Tesla.Adapter.Finch, name: MyFinch})
  end
end

  
    
  
  Streaming Response Body


If adapter supports it, you can pass a response: :stream option to return
response body as a Stream
defmodule OpenAI do
  def client(token) do
    middleware = [
      {Tesla.Middleware.BaseUrl, "https://api.openai.com/v1"},
      {Tesla.Middleware.BearerAuth, token: token},
      {Tesla.Middleware.JSON, decode_content_types: ["text/event-stream"]},
      {Tesla.Middleware.SSE, only: :data}
    ]
    Tesla.client(middleware, {Tesla.Adapter.Finch, name: MyFinch})
  end

  def completion(client, prompt) do
    data = %{
      model: "gpt-3.5-turbo",
      messages: [%{role: "user", content: prompt}],
      stream: true
    }
    Tesla.post(client, "/chat/completions", data, opts: [adapter: [response: :stream]])
  end
end

client = OpenAI.new("<token>")
{:ok, env} = OpenAI.completion(client, "What is the meaning of life?")
env.body |> Stream.each(fn chunk -> IO.inspect(chunk) end)

  
    
  
  Middleware



  
    
  
  Custom middleware


defmodule Tesla.Middleware.MyCustomMiddleware do
  @moduledoc """
  Short description what it does

  Longer description, including e.g. additional dependencies.

  ### Options

  - `:list` - all possible options
  - `:with` - their default values

  ### Examples

      client = Tesla.client([{Tesla.Middleware.MyCustomMiddleware, with: value}])
  """

  @behaviour Tesla.Middleware

  @impl Tesla.Middleware
  def call(env, next, options) do
    with %Tesla.Env{} = env <- preprocess(env) do
      env
      |> Tesla.run(next)
      |> postprocess()
    end
  end

  defp preprocess(env) do
    env
  end

  defp postprocess({:ok, env}) do
    {:ok, env}
  end

  def postprocess({:error, reason}) do
    {:error, reason}
  end
end

  
    
  
  Adapter



  
    
  
  Custom adapter


defmodule Tesla.Adapter.MyCustomAdapter do
  @behaviour Tesla.Adapter

  @impl Tesla.Adapter
  def run(env, opts) do
    # do something
  end
end

Migrate from v0 to v1
    

This is a list of all breaking changes.
Version 1.0 has been released, try it today!
  defp deps do
    [
      {:tesla, "1.0.0"}
    ]
  end
Any other breaking change not on this list is considered a bug - in you find one please create a new issue.

  
    
  
  Returning Tuple Result from HTTP Functions


get(..), post(..), etc. now return {:ok, Tesla.Env} | {:error, reason} (#177)
In 0.x all http functions returned either Tesla.Env or raised an error.
In 1.0 these functions return ok/error tuples. The old behaviour can be achieved with the new ! (bang) functions: get!(...), post!(...), etc.
case MyApi.get("/") do
  {:ok, %Tesla.Env{status: 200}} -> # ok response
  {:ok, %Tesla.Env{status: 500}} -> # server error
  {:error, reason} -> # connection & other errors
end

  
    
  
  Dropped aliases support (#159)


Use full module name for middleware and adapters.
# middleware
-  plug :json
+  plug Tesla.Middleware.JSON

# adapter
-  adapter :hackney
+  adapter Tesla.Adapter.Hackney

# config
-  config :tesla, adapter: :mock
+  config :tesla, adapter: Tesla.Mock

  
    
  
  Dropped local middleware/adapter functions (#171)


Extract functionality into separate module.
 defmodule MyClient do
-  plug :some_local_fun
-
-  def some_local_fun(env, next) do
     # implementation
-  end
 end

+defmodule ProperlyNamedMiddleware do
+  @behaviour Tesla.Middleware
+  def call(env, next, _opts) do
     # implementation
+  end
+end

 defmodule MyClient do
+  plug ProperlyNamedMiddleware
 end

  
    
  
  Dropped client as function (#176)


This is very unlikely, but... if you hacked around with custom functions as client (the first argument) you need to stop.
See Tesla.client/2 instead.

  
    
  
  Headers are now a list (#160)


In 0.x env.headers are a map(binary => binary).
In 1.x env.headers are a [{binary, binary}].
This change also applies to middleware headers.
Setting a header
-  env
-  |> Map.update!(&Map.put(&1.headers, "name", "value"))

+  env
+  |> Tesla.put_header("name", "value")
Getting a header
-  env.headers["cookie"]
+  Tesla.get_header(env, "cookie") # => "secret"
+  Tesla.get_headers(env, "cookie") # => ["secret", "token", "and more"]


-  case env.headers do
-    %{"server" => server} -> ...
-    _ -> ...
-  end
+  case Tesla.get_header(env, "server") do
+    nil -> ...
+    server ->
+  end
There are five new functions to deal with headers:
	Tesla.get_header(env, name) :: binary | nil - Get first header with given name

	Tesla.get_headers(env, name) :: [binary] - Get all headers values with given name
	Tesla.put_header(env, name, value) - Set header with given name and value. Existing header with the same name will be overwritten.
	Tesla.put_headers(env, list) - Add headers to the end of env.headers. Does not make the headers unique.
	Tesla.delete_header(env, name) - Delete all headers with given name


  
    
  
  Dropped support for Elixir 1.3 (#164)


Tesla 1.0 works only with Elixir 1.4 or newer

  
    
  
  Adapter options need to be wrapped in :adapter key:


- MyClient.get("/", opts: [recv_timeout: 30_000])
+ MyClient.get("/", opts: [adapter: [recv_timeout: 30_000]])

  
    
  
  DebugLogger merged into Logger (#150)


Debugging request and response details has been merged into a single Logger middleware. See Tesla.Middleware.Logger documentation for more information.
  defmodule MyClient do
    use Tesla

    plug Tesla.Middleware.Logger
-   plug Tesla.Middleware.DebugLogger
  end

  
    
  
  Jason is the new default JSON library (#175)


The Tesla.Middleware.JSON now requires jason by default. If you want to keep using poison you will have to set :engine option - see documentation for details.

Migrate away from v1 Macro
    

We encourage users to contribute to this guide to help others migrate away from
the v1 macro syntax. Every case is different, so we can't provide a
one-size-fits-all solution, but we can provide a guide to help you migrate your
codebase.
Please share your learnings and suggestions in the Migrating away from v1 Macro GitHub Discussion.
	Find all the modules that use use Tesla
defmodule MyApp.MyTeslaClient do
  use Tesla # <- this line
end

	Remove use Tesla
 - defmodule MyApp.MyTeslaClient do
 -   use Tesla
 - end
 + defmodule MyApp.MyTeslaClient do
 + end

	Find all the plug macro calls:
defmodule MyApp.MyTeslaClient do
  plug Tesla.Middleware.KeepRequest # <- this line
  plug Tesla.Middleware.PathParams # <- this line
  plug Tesla.Middleware.JSON # <- this line
end

	Move all the plug macro calls to a function that returns the middleware.
defmodule MyApp.MyTeslaClient do
- plug Tesla.Middleware.KeepRequest
- plug Tesla.Middleware.PathParams
- plug Tesla.Middleware.JSON
+
+ def middleware do
+   [Tesla.Middleware.KeepRequest, Tesla.Middleware.PathParams, Tesla.Middleware.JSON]
+ end
end

	Find all the adapter macro calls:
defmodule MyApp.MyTeslaClient do
  adapter Tesla.Adapter.Hackney # <- this line
  adapter fn env -> # <- or this line
  end
end

	Create a adapter/0 function that returns the adapter to use for that given
module, or however you prefer to configure the adapter used. Please refer to
the Adapter Explanation documentation for more
information.
Context Matters
This step is probably the most important one. The context in which the
adapter is used matters a lot. Please be careful with this step, and test
your changes thoroughly.
defmodule MyApp.MyTeslaClient do
- adapter Tesla.Adapter.Hackney
+ defp adapter do
+   # if the value is `nil`, the default global Tesla adapter will be used
+   # which is the existing behavior.
+   :my_app
+   |> Application.get_env(__MODULE__, [])
+   |> Keyword.get(:adapter)
+ end
end

	Create a client/0 function that returns a Tesla.Client struct with the
middleware and adapter. Please refer to the Client Explanation
documentation for more information.
defmodule MyApp.MyTeslaClient do
  def client do
    Tesla.client(middleware(), adapter())
  end

  defp middleware do
    [Tesla.Middleware.KeepRequest, Tesla.Middleware.PathParams, Tesla.Middleware.JSON]
  end

  defp adapter do
    :my_app
    |> Application.get_env(__MODULE__, [])
    |> Keyword.get(:adapter)
  end
end

	Replace all the Tesla.get/2, Tesla.post/2, etc. to receive the client
as an argument.
defmodule MyApp.MyTeslaClient do
  def do_something do
-   get("/endpoint")
+   Tesla.get!(client(), "/endpoint")
  end
end

Test Using Mox
    

To mock HTTP requests in your tests using Mox with the Tesla HTTP client,
follow these steps:

  
    
  
  1. Define a Mock Adapter


First, define a mock adapter that implements the Tesla.Adapter behaviour. This
adapter will intercept HTTP requests during testing.
Create a file at test/support/mocks.ex:
# test/support/mocks.ex
Mox.defmock(MyApp.MockAdapter, for: Tesla.Adapter)

  
    
  
  2. Configure the Mock Adapter for Tests


In your config/test.exs file, configure Tesla to use the mock adapter you
just defined:
# config/test.exs
config :tesla, adapter: MyApp.MockAdapter
If you are not using the global adapter configuration, ensure that your Tesla
client modules are configured to use MyApp.MockAdapter during tests.

  
    
  
  3. Set Up Mocking in Your Tests


Create a test module, for example test/demo_test.exs, and set up Mox to
define expectations and verify them:
defmodule MyApp.FeatureTest do
  use ExUnit.Case, async: true

  import Mox
  import Tesla.Test

  setup :set_mox_from_context
  setup :verify_on_exit!

  test "example test" do
    #--------- Given - Stubs and Preconditions
    # Expect a single HTTP request to be made and return a JSON response
    expect_tesla_call(
      times: 1,
      returns: json(%Tesla.Env{status: 200}, %{id: 1})
    )

    #--------- When - Run the code under test
    # Make the HTTP request using Tesla
    # Mimic a use case where we create a user
    assert :ok = create_user!(%{username: "johndoe"})

    #--------- Then - Assert postconditions
    # Verify that the HTTP request was made and matches the expected parameters
    assert_received_tesla_call(env, [])
    assert env.status == 200
    assert env.method == :post
    assert env.url == "https://acme.com/users"
    # ...

    # Or you can verify the entire `t:Tesla.Env.t/0` using something like this:
    assert_tesla_env(env, %Tesla.Env{
      method: :post,
      url: "https://acme.com/users",
      body: %{username: "johndoe"},
      status: 200,
    })

    # Verify that the mailbox is empty, indicating no additional requests were
    # made and all messages have been processed
    assert_tesla_empty_mailbox()
  end

  defp create_user!(body) do
    # ...
    Tesla.post!("https://acme.com/users", body)
    # ...
    :ok
  end
end

  
    
  
  4. Run Your Tests


When you run your tests with mix test, all HTTP requests made by Tesla will
be intercepted by MyApp.MockAdapter, and responses will be provided based
on your Mox expectations.

A HTTP toolkit for building API clients using middlewares.

  
    
  
  Building API client


Use Tesla.client/2 to build a client with the given middleware and adapter.

  
    
  
  Examples


defmodule ExampleApi do
  def client do
    Tesla.client([
      {Tesla.Middleware.BaseUrl, "http://api.example.com"},
      Tesla.Middleware.JSON
    ])
  end

  def fetch_data(client) do
    Tesla.get(client, "/data")
  end
end
Now you can use ExampleApi.client/0 to make requests to the API.
client = ExampleApi.client()
ExampleApi.fetch_data(client)

  
    
  
  Direct usage


It is also possible to do request directly with Tesla module.
Tesla.get("https://example.com")

  
    
  
  Default adapter


By default Tesla is using Tesla.Adapter.Httpc, because :httpc is
included in Erlang/OTP and does not require installation of any additional
dependency. It can be changed globally with config:
config :tesla, :adapter, Tesla.Adapter.Mint

Tesla.Env 
    



      
This module defines a Tesla.Env.t/0 struct that stores all data related to request/response.

  
    
  
  Fields


	:method - method of request. Example: :get
	:url - request url. Example: "https://www.google.com"
	:query - list of query params.
Example: [{"param", "value"}] will be translated to ?params=value.
Note: query params passed in url (e.g. "/get?param=value") are not parsed to query field.
	:headers - list of request/response headers.
Example: [{"content-type", "application/json"}].
Note: request headers are overridden by response headers when adapter is called.
	:body - request/response body.
Note: request body is overridden by response body when adapter is called.
	:status - response status. Example: 200
	:opts - list of options. Example: [adapter: [recv_timeout: 30_000]]

Tesla.Mock 
    



      
Mock adapter for better testing.

  
    
  
  Setup


# config/test.exs
config :tesla, adapter: Tesla.Mock

# in case MyClient defines specific adapter with `adapter SpecificAdapter`
config :tesla, MyClient, adapter: Tesla.Mock

  
    
  
  Examples


defmodule MyAppTest do
  use ExUnit.Case

  setup do
    Tesla.Mock.mock(fn
      %{method: :get} ->
        %Tesla.Env{status: 200, body: "hello"}
    end)

    :ok
  end

  test "list things" do
    assert {:ok, env} = MyApp.get("...")
    assert env.status == 200
    assert env.body == "hello"
  end
end

  
    
  
  Setting up mocks


# Match on method & url and return whole Tesla.Env
Tesla.Mock.mock(fn
  %{method: :get, url: "http://example.com/list"} ->
    %Tesla.Env{status: 200, body: "hello"}
end)

# You can use any logic required
Tesla.Mock.mock(fn env ->
  case env.url do
    "http://example.com/list" ->
      %Tesla.Env{status: 200, body: "ok!"}

    _ ->
      %Tesla.Env{status: 404, body: "NotFound"}
  end
end)


# mock will also accept short version of response
# in the form of {status, headers, body}
Tesla.Mock.mock(fn
  %{method: :post} -> {201, %{}, %{id: 42}}
end)

# mock will also accept error tuples in the form
# of {:error, reason}
Tesla.Mock.mock(fn
  %{method: :post} -> {:error, :timeout}
end)

  
    
  
  Global mocks


By default, mocks are bound to the current process,
i.e. the process running a single test case.
This design allows proper isolation between test cases
and make testing in parallel (async: true) possible.
While this style is recommended, there is one drawback:
if Tesla client is called from different process
it will not use the setup mock.
To solve this issue it is possible to setup a global mock
using mock_global/1 function.
defmodule MyTest do
  use ExUnit.Case, async: false # must be false!

  setup_all do
    Tesla.Mock.mock_global fn
      env -> # ...
    end

    :ok
  end

  # ...
end
WARNING: Using global mocks may affect tests with local mock
(because of fallback to global mock in case local one is not found)

Tesla.Multipart 
    



      
Multipart functionality.

  
    
  
  Examples


mp =
  Multipart.new()
  |> Multipart.add_content_type_param("charset=utf-8")
  |> Multipart.add_field("field1", "foo")
  |> Multipart.add_field("field2", "bar",
    headers: [{"content-id", "1"}, {"content-type", "text/plain"}]
  )
  |> Multipart.add_file("test/tesla/multipart_test_file.sh")
  |> Multipart.add_file("test/tesla/multipart_test_file.sh", name: "foobar")
  |> Multipart.add_file_content("sample file content", "sample.txt")

response = client.post(url, mp)

Tesla.Test 
    



      
Provides utilities for testing Tesla-based HTTP clients.

Tesla.Adapter behaviour
    



      
The adapter specification.
Adapter is a module that denormalize request data stored in Tesla.Env in order to make
request with lower level http client (e.g. :httpc or :hackney) and normalize response data
in order to store it back to Tesla.Env. It has to implement Tesla.Adapter.call/2.

  
    
  
  Writing custom adapter


Create a module implementing Tesla.Adapter.call/2.
See Tesla.Adapter.call/2 for details.

  
    
  
  Examples


defmodule MyProject.CustomAdapter do
  alias Tesla.Multipart

  @behaviour Tesla.Adapter

  @override_defaults [follow_redirect: false]

  @impl Tesla.Adapter
  def call(env, opts) do
    opts = Tesla.Adapter.opts(@override_defaults, env, opts)

    with {:ok, {status, headers, body}} <- request(env.method, env.body, env.headers, opts) do
      {:ok, normalize_response(env, status, headers, body)}
    end
  end

  defp request(_method, %Stream{}, _headers, _opts) do
    {:error, "stream not supported by adapter"}
  end

  defp request(_method, %Multipart{}, _headers, _opts) do
    {:error, "multipart not supported by adapter"}
  end

  defp request(method, body, headers, opts) do
    :lower_level_http.request(method, body, denormalize_headers(headers), opts)
  end

  defp denormalize_headers(headers), do: ...
  defp normalize_response(env, status, headers, body), do: %Tesla.Env{env | ...}
end

The middleware specification.
Middleware is an extension of basic Tesla functionality. It is a module that must
implement Tesla.Middleware.call/3.

  
    
  
  Middleware options


Options can be passed to middleware inside tuple in case of dynamic middleware
(Tesla.client/1):
Tesla.client([{Tesla.Middleware.BaseUrl, "https://example.com"}])

  
    
  
  Ordering


The order in which middleware is defined matters. Note that the order when sending the request
matches the order the middleware was defined in, but the order when receiving the response
is reversed.
For example, Tesla.Middleware.DecompressResponse must come after Tesla.Middleware.JSON,
otherwise the response isn't decompressed before it reaches the JSON parser.

  
    
  
  Writing custom middleware


Writing custom middleware is as simple as creating a module implementing Tesla.Middleware.call/3.
See Tesla.Middleware.call/3 for details.

  
    
  
  Examples


defmodule MyProject.InspectHeadersMiddleware do
  @behaviour Tesla.Middleware

  @impl true
  def call(env, next, _options) do
    IO.inspect(env.headers)

    with {:ok, env} <- Tesla.run(env, next) do
      IO.inspect(env.headers)
      {:ok, env}
    end
  end
end

Tesla.Adapter.Finch 
    



      
Adapter for finch.
Remember to add {:finch, "~> 0.14.0"} to dependencies. Also, you need to
recompile tesla after adding the :finch dependency:
mix deps.clean tesla
mix compile


  
    
  
  Examples


In order to use Finch, you must start it and provide a :name. For example,
in your supervision tree:
children = [
  {Finch, name: MyFinch}
]
You must provide the same name to this adapter:
# set globally in config/config.exs
config :tesla, :adapter, {Tesla.Adapter.Finch, name: MyFinch}

# set per module
defmodule MyClient do
  def client do
    Tesla.client([], {Tesla.Adapter.Finch, name: MyFinch})
  end
end

  
    
  
  Adapter specific options


	:name - The :name provided to Finch (required).


  
    
  
  Finch options


	:pool_timeout - This timeout is applied when a connection is checked
out from the pool. Default value is 5_000.

	:receive_timeout - The maximum time to wait for a response before
returning an error. Default value is 15_000.

Adapter for gun.
Remember to add {:gun, "~> 1.3"}, {:idna, "~> 6.0"} and {:castore, "~> 0.1"} to dependencies.
In version 1.3 gun sends host header with port. Fixed in master branch.
Also, you need to recompile tesla after adding :gun dependency:
mix deps.clean tesla
mix deps.compile tesla


  
    
  
  Examples


# set globally in config/config.exs
config :tesla, :adapter, Tesla.Adapter.Gun

# set per module
defmodule MyClient do
  def client do
    Tesla.client([], Tesla.Adapter.Gun)
  end
end

  
    
  
  Adapter specific options


	:timeout - Time, while process, will wait for gun messages.

	:body_as - What will be returned in %Tesla.Env{} body key. Possible values:
	:plain - as binary (default).
	:stream - as stream.
  If you don't want to close connection (because you want to reuse it later)
  pass close_conn: false in adapter opts.
	:chunks - as chunks.
  You can get response body in chunks using Tesla.Adapter.Gun.read_chunk/3 function.

  Processing of the chunks and checking body size must be done by yourself.
  Example of processing function is in test/tesla/adapter/gun_test.exs - Tesla.Adapter.GunTest.read_body/4.
  If you don't need connection later don't forget to close it with Tesla.Adapter.Gun.close/1.

	:max_body - Max response body size in bytes.
  Works only with body_as: :plain, with other settings you need to check response body size by yourself.

	:conn - Opened connection pid with gun. Is used for reusing gun connections.

	:close_conn - Close connection or not after receiving full response body.
  Is used for reusing gun connections. Defaults to true.

	:certificates_verification - Add SSL certificates verification.
  erlang-certifi
  ssl_verify_fun.erl

	:proxy - Proxy for requests.
  Socks proxy are supported from gun >= 2.0.
  Examples: {'localhost', 1234}, {{127, 0, 0, 1}, 1234}, {:socks5, 'localhost', 1234}.
NOTE: By default GUN uses TLS as transport if the specified port is 443,
  if TLS is required for proxy connection on another port please specify transport
  using the Gun options below otherwise tcp will be used.

	:proxy_auth - Auth to be passed along with the proxy opt.
  Supports Basic auth for regular and Socks proxy.
  Format: {proxy_username, proxy_password}.



  
    
  
  Gun options


	:connect_timeout - Connection timeout.

	:http_opts - Options specific to the HTTP protocol.

	:http2_opts -  Options specific to the HTTP/2 protocol.

	:protocols - Ordered list of preferred protocols.
  Defaults: [:http2, :http]- for :tls, [:http] - for :tcp.

	:trace - Whether to enable dbg tracing of the connection process.
  Should only be used during debugging. Default: false.

	:transport - Whether to use TLS or plain TCP.
  The default varies depending on the port used.
  Port 443 defaults to tls. All other ports default to tcp.

	:transport_opts - Transport options.
  They are TCP options or TLS options depending on the selected transport.
  Default: []. Gun version: 1.3.

	:tls_opts - TLS transport options.
  Default: []. Gun from master branch.

	:tcp_opts - TCP transport options.
  Default: []. Gun from master branch.

	:socks_opts - Options for socks.
  Default: []. Gun from master branch.

	:ws_opts - Options specific to the Websocket protocol. Default: %{}.
	:compress - Whether to enable permessage-deflate compression.
  This does not guarantee that compression will be used as it is the server
  that ultimately decides. Defaults to false.

	:protocols - A non-empty list enables Websocket protocol negotiation.
  The list of protocols will be sent in the sec-websocket-protocol request header.
  The handler module interface is currently undocumented and must be set to gun_ws_h.

Tesla.Adapter.Hackney 
    



      
Adapter for hackney.
Remember to add {:hackney, "~> 1.13"} to dependencies (and :hackney to applications in mix.exs)
Also, you need to recompile tesla after adding :hackney dependency:
mix deps.clean tesla
mix deps.compile tesla


  
    
  
  Examples


# set globally in config/config.exs
config :tesla, :adapter, Tesla.Adapter.Hackney

# set per module
defmodule MyClient do
  def client do
    Tesla.client([], Tesla.Adapter.Hackney)
  end
end

  
    
  
  Adapter specific options


	:max_body - Max response body size in bytes. Actual response may be bigger because hackney stops after the last chunk that surpasses :max_body.

Adapter for httpc.
This is the default adapter.
NOTE Tesla overrides default autoredirect value with false to ensure
consistency between adapters

Tesla.Adapter.Ibrowse 
    



      
Adapter for ibrowse.
Remember to add {:ibrowse, "~> 4.2"} to dependencies (and :ibrowse to applications in mix.exs)
Also, you need to recompile tesla after adding :ibrowse dependency:
mix deps.clean tesla
mix deps.compile tesla

  
    
  
  Examples


# set globally in config/config.exs
config :tesla, :adapter, Tesla.Adapter.Ibrowse

# set per module
defmodule MyClient do
  def client do
    Tesla.client([], Tesla.Adapter.Ibrowse)
  end
end

Tesla.Adapter.Mint 
    



      
Adapter for mint.
NOTE: The minimum supported Elixir version for mint is 1.5.0
Remember to add {:mint, "~> 1.0"} and {:castore, "~> 0.1"} to dependencies.
Also, you need to recompile tesla after adding :mint dependency:
mix deps.clean tesla
mix deps.compile tesla


  
    
  
  Examples


# set globally in config/config.exs
config :tesla, :adapter, Tesla.Adapter.Mint
# set per module
defmodule MyClient do
  def client do
    Tesla.client([], Tesla.Adapter.Mint)
  end
end

# set global custom cacertfile
config :tesla, adapter: {Tesla.Adapter.Mint, cacert: ["path_to_cacert"]}

  
    
  
  Adapter specific options:


	:timeout - Time in milliseconds, while process, will wait for mint messages. Defaults to 2_000.
	:body_as - What will be returned in %Tesla.Env{} body key. Possible values - :plain, :stream, :chunks. Defaults to :plain.	:plain - as binary.
	:stream - as stream. If you don't want to close connection (because you want to reuse it later) pass close_conn: false in adapter opts.
	:chunks - as chunks. You can get response body in chunks using Tesla.Adapter.Mint.read_chunk/3 function.
Processing of the chunks and checking body size must be done by yourself. Example of processing function is in test/tesla/adapter/mint_test.exs - Tesla.Adapter.MintTest.read_body/4. If you don't need connection later don't forget to close it with Tesla.Adapter.Mint.close/1.


	:max_body - Max response body size in bytes. Works only with body_as: :plain, with other settings you need to check response body size by yourself.
	:conn - Opened connection with mint. Is used for reusing mint connections.
	:original - Original host with port, for which reused connection was open. Needed for Tesla.Middleware.FollowRedirects. Otherwise adapter will use connection for another open host.
	:close_conn - Close connection or not after receiving full response body. Is used for reusing mint connections. Defaults to true.
	:proxy - Proxy settings. E.g.: {:http, "localhost", 8888, []}, {:http, "127.0.0.1", 8888, []}
	:transport_opts - Keyword list of HTTP or HTTPS options passed into :gen_tcp or :ssl respectively by mint. See mint's docs on transport_opts.

Tesla.Middleware.BaseUrl 
    



      
Set base URL for all requests.
The base URL will be prepended to request path/URL only
if it does not include http(s).

  
    
  
  Examples


defmodule MyClient do
  def client do
    Tesla.client([
      {Tesla.Middleware.BaseUrl, "https://example.com/foo"}
    ])
  end
end

client = MyClient.client()

Tesla.get(client, "/path")
# equals to GET https://example.com/foo/path

Tesla.get(client, "path")
# equals to GET https://example.com/foo/path

Tesla.get(client, "")
# equals to GET https://example.com/foo

Tesla.get(client, "http://example.com/bar")
# equals to GET http://example.com/bar

Tesla.Middleware.BasicAuth 
    



      
Basic authentication middleware.
Wiki on the topic

  
    
  
  Examples


defmodule MyClient do
  def client(username, password, opts \ %{}) do
    Tesla.client([
      {Tesla.Middleware.BasicAuth,
        Map.merge(%{username: username, password: password}, opts)}
    ])
  end
end

  
    
  
  Options


	:username - username (defaults to "")
	:password - password (defaults to "")

Tesla.Middleware.BearerAuth 
    



      
Bearer authentication middleware.
Adds a {"authorization", "Bearer <token>"} header.

  
    
  
  Examples


defmodule MyClient do
  def new(token) do
    Tesla.client([
      {Tesla.Middleware.BearerAuth, token: token}
    ])
  end
end

  
    
  
  Options


	:token - token (defaults to "")

Tesla.Middleware.CompressRequest 
    



      
Only compress request.
See Tesla.Middleware.Compression for options.

Tesla.Middleware.Compression 
    



      
Compress requests and decompress responses.
Supports "gzip" and "deflate" encodings using Erlang's built-in :zlib module.

  
    
  
  Examples


defmodule MyClient do
  def client do
    Tesla.client([
      {Tesla.Middleware.Compression, format: "gzip"}
    ])
  end
end

  
    
  
  Options


	:format - request compression format, "gzip" (default) or "deflate"

Tesla.Middleware.DecodeJson 
    



      
Decodes response body as JSON.
Only decodes the body if the Content-Type header suggests
that the body is JSON.

Tesla.Middleware.DecodeRels 
    



      
Decode Link Hypermedia HTTP header into opts[:rels] field in response.

  
    
  
  Examples


defmodule MyClient do
  def client do
    Tesla.client([Tesla.Middleware.DecodeRels])
  end
end

client = MyClient.client()

env = Tesla.get(client, "/...")

env.opts[:rels]
# => %{"Next" => "http://...", "Prev" => "..."}

Only decompress response.
See Tesla.Middleware.Compression for options.

Tesla.Middleware.DigestAuth 
    



      
Digest access authentication middleware.
Wiki on the topic
NOTE: Currently the implementation is incomplete and works only for MD5 algorithm
and auth "quality of protection" (qop).

  
    
  
  Examples


defmodule MyClient do
  def client(username, password, opts \ %{}) do
    Tesla.client([
      {Tesla.Middleware.DigestAuth, Map.merge(%{username: username, password: password}, opts)}
    ])
  end
end

  
    
  
  Options


	:username - username (defaults to "")
	:password - password (defaults to "")
	:cnonce_fn - custom function generating client nonce (defaults to &Tesla.Middleware.DigestAuth.cnonce/0)
	:nc - nonce counter (defaults to "00000000")

Tesla.Middleware.EncodeJson 
    



      
Encodes request body as JSON.

Tesla.Middleware.FollowRedirects 
    



      
Follow HTTP 3xx redirects.

  
    
  
  Examples


defmodule MyClient do
  def client do
  # defaults to 5
    Tesla.client([
      {Tesla.Middleware.FollowRedirects, max_redirects: 3}
    ])
  end
end

  
    
  
  Options


	:max_redirects - limit number of redirects (default: 5)

Tesla.Middleware.FormUrlencoded 
    



      
Send request body as application/x-www-form-urlencoded.
Performs encoding of body from a Map such as %{"foo" => "bar"} into
URL-encoded data.
Performs decoding of the response into a map when urlencoded and content-type
is application/x-www-form-urlencoded, so "foo=bar" becomes
%{"foo" => "bar"}.

  
    
  
  Examples


defmodule Myclient do
  def client do
    Tesla.client([
      {Tesla.Middleware.FormUrlencoded,
        encode: &Plug.Conn.Query.encode/1,
        decode: &Plug.Conn.Query.decode/1}
    ])
  end
end

client = Myclient.client()
Myclient.post(client, "/url", %{key: :value})

  
    
  
  Options


	:decode - decoding function, defaults to URI.decode_query/1
	:encode - encoding function, defaults to URI.encode_query/1


  
    
  
  Nested Maps


Natively, nested maps are not supported in the body, so
%{"foo" => %{"bar" => "baz"}} won't be encoded and raise an error.
Support for this specific case is obtained by configuring the middleware to
encode (and decode) with Plug.Conn.Query
defmodule Myclient do
  def client do
    Tesla.client([
      {Tesla.Middleware.FormUrlencoded,
        encode: &Plug.Conn.Query.encode/1,
        decode: &Plug.Conn.Query.decode/1}
    ])
  end
end

client = Myclient.client()
Myclient.post